

## A13. Grammar

Below is a recapitulation of the grammar that was given throughout the earlier part of this appendix. It has exactly the same content, but is in a different order.

The grammar has undefined terminal symbols *integer-constant*, *character-constant*, *floating-constant*, *identifier*, *string*, and *enumeration-constant*; the *typewriter* style words and symbols are terminals given literally. This grammar can be transformed mechanically into input acceptable to an automatic parser-generator. Besides adding whatever syntactic marking is used to indicate alternatives in productions, it is necessary to expand the "one of" constructions, and (depending on the rules of the parser-generator) to duplicate each production with an *opt* symbol, once with the symbol and once without. With one further change, namely deleting the production *typedef-name: identifier* and making *typedef-name* a terminal symbol, this grammar is acceptable to the YACC parser-generator. It has only one conflict, generated by the *if-else* ambiguity.

*translation-unit:*  
*external-declaration*  
*translation-unit external-declaration*

*external-declaration:*  
*function-definition*  
*declaration*

*function-definition:*  
*declaration-specifiers*<sub>opt</sub> *declarator declaration-list*<sub>opt</sub> *compound-statement*

*declaration:*  
*declaration-specifiers init-declarator-list*<sub>opt</sub> ;

*declaration-list:*  
*declaration*  
*declaration-list declaration*

*declaration-specifiers:*  
*storage-class-specifier declaration-specifiers*<sub>opt</sub>  
*type-specifier declaration-specifiers*<sub>opt</sub>  
*type-qualifier declaration-specifiers*<sub>opt</sub>

*storage-class-specifier:* one of  
*auto register static extern typedef*

*type-specifier:* one of  
*void char short int long float double signed*  
*unsigned struct-or-union-specifier enum-specifier typedef-name*

*type-qualifier:* one of  
*const volatile*

*struct-or-union-specifier:*  
*struct-or-union identifier*<sub>opt</sub> { *struct-declaration-list* }  
*struct-or-union identifier*

*struct-or-union:* one of  
*struct union*

*struct-declaration-list:*  
*struct-declaration*  
*struct-declaration-list struct-declaration*

*init-declarator-list:*  
*init-declarator*  
*init-declarator-list , init-declarator*

*init-declarator:*  
*declarator*  
*declarator = initializer*

*struct-declaration:*  
*specifier-qualifier-list struct-declarator-list ;*

*specifier-qualifier-list:*  
*type-specifier specifier-qualifier-list*<sub>opt</sub>  
*type-qualifier specifier-qualifier-list*<sub>opt</sub>

*struct-declarator-list:*  
*struct-declarator*  
*struct-declarator-list , struct-declarator*

*struct-declarator:*  
*declarator*  
*declarator*<sub>opt</sub> : *constant-expression*

*enum-specifier:*  
*enum identifier*<sub>opt</sub> { *enumerator-list* }  
*enum identifier*

*enumerator-list:*  
*enumerator*  
*enumerator-list , enumerator*

*enumerator:*  
*identifier*  
*identifier = constant-expression*

*declarator:*  
*pointer*<sub>opt</sub> *direct-declarator*

*direct-declarator:*  
*identifier*  
 ( *declarator* )  
*direct-declarator* [ *constant-expression*<sub>opt</sub> ]  
*direct-declarator* ( *parameter-type-list* )  
*direct-declarator* ( *identifier-list*<sub>opt</sub> )

*pointer:*  
 \* *type-qualifier-list*<sub>opt</sub>  
 \* *type-qualifier-list*<sub>opt</sub> *pointer*

*type-qualifier-list:*  
*type-qualifier*  
*type-qualifier-list type-qualifier*

*parameter-type-list:*  
*parameter-list*  
*parameter-list , ...*

*parameter-list:*  
*parameter-declaration*  
*parameter-list , parameter-declaration*

*parameter-declaration:*  
*declaration-specifiers declarator*  
*declaration-specifiers abstract-declarator<sub>opt</sub>*

*identifier-list:*  
*identifier*  
*identifier-list , identifier*

*initializer:*  
*assignment-expression*  
*{ initializer-list }*  
*{ initializer-list , }*

*initializer-list:*  
*initializer*  
*initializer-list , initializer*

*type-name:*  
*specifier-qualifier-list abstract-declarator<sub>opt</sub>*

*abstract-declarator:*  
*pointer*  
*pointer<sub>opt</sub> direct-abstract-declarator*

*direct-abstract-declarator:*  
*( abstract-declarator )*  
*direct-abstract-declarator<sub>opt</sub> [ constant-expression<sub>opt</sub> ]*  
*direct-abstract-declarator<sub>opt</sub> ( parameter-type-list<sub>opt</sub> )*

*typedef-name:*  
*identifier*

*statement:*  
*labeled-statement*  
*expression-statement*  
*compound-statement*  
*selection-statement*  
*iteration-statement*  
*jump-statement*

*labeled-statement:*  
*identifier : statement*  
*case constant-expression : statement*  
*default : statement*

*expression-statement:*  
*expression<sub>opt</sub> ;*

*compound-statement:*  
*{ declaration-list<sub>opt</sub> statement-list<sub>opt</sub> }*

*statement-list:*  
*statement*  
*statement-list statement*

*selection-statement:*  
*if ( expression ) statement*  
*if ( expression ) statement else statement*  
*switch ( expression ) statement*

*iteration-statement:*  
*while ( expression ) statement*  
*do statement while ( expression ) ;*  
*for ( expression<sub>opt</sub> ; expression<sub>opt</sub> ; expression<sub>opt</sub> ) statement*

*jump-statement:*  
*goto identifier ;*  
*continue ;*  
*break ;*  
*return expression<sub>opt</sub> ;*

*expression:*  
*assignment-expression*  
*expression , assignment-expression*

*assignment-expression:*  
*conditional-expression*  
*unary-expression assignment-operator assignment-expression*

*assignment-operator:* one of  
 = \* = / = % = + = - = < = > = & = ^ = ! =

*conditional-expression:*  
*logical-OR-expression*  
*logical-OR-expression ? expression : conditional-expression*

*constant-expression:*  
*conditional-expression*

*logical-OR-expression:*  
*logical-AND-expression*  
*logical-OR-expression || logical-AND-expression*

*logical-AND-expression:*  
*inclusive-OR-expression*  
*logical-AND-expression && inclusive-OR-expression*

*inclusive-OR-expression:*  
*exclusive-OR-expression*  
*inclusive-OR-expression | exclusive-OR-expression*

*exclusive-OR-expression:*  
*AND-expression*  
*exclusive-OR-expression ^ AND-expression*

*AND-expression:*  
*equality-expression*  
*AND-expression & equality-expression*

*equality-expression:*  
*relational-expression*  
*equality-expression == relational-expression*  
*equality-expression != relational-expression*

*relational-expression:*  
*shift-expression*  
*relational-expression < shift-expression*  
*relational-expression > shift-expression*  
*relational-expression <= shift-expression*  
*relational-expression >= shift-expression*

*shift-expression*:

- additive-expression*
- shift-expression* << *additive-expression*
- shift-expression* >> *additive-expression*

*additive-expression*:

- multiplicative-expression*
- additive-expression* + *multiplicative-expression*
- additive-expression* - *multiplicative-expression*

*multiplicative-expression*:

- cast-expression*
- multiplicative-expression* \* *cast-expression*
- multiplicative-expression* / *cast-expression*
- multiplicative-expression* % *cast-expression*

*cast-expression*:

- unary-expression*
- ( *type-name* ) *cast-expression*

*unary-expression*:

- postfix-expression*
- ++ *unary-expression*
- *unary-expression*
- unary-operator* *cast-expression*
- sizeof *unary-expression*
- sizeof ( *type-name* )

*unary-operator*: one of

& \* + - ~ !

*postfix-expression*:

- primary-expression*
- postfix-expression* [ *expression* ]
- postfix-expression* ( *argument-expression-list*<sub>opt</sub> )
- postfix-expression* . *identifier*
- postfix-expression* -> *identifier*
- postfix-expression* ++
- postfix-expression* --

*primary-expression*:

- identifier*
- constant*
- string*
- ( *expression* )

*argument-expression-list*:

- assignment-expression*
- argument-expression-list* , *assignment-expression*

*constant*:

- integer-constant*
- character-constant*
- floating-constant*
- enumeration-constant*

The following grammar for the preprocessor summarizes the structure of control lines, but is not suitable for mechanized parsing. It includes the symbol *text*, which means ordinary program text, non-conditional preprocessor control lines, or complete preprocessor conditional constructions.

*control-line*:

- # *define identifier token-sequence*
- # *define identifier*( *identifier* , ... , *identifier* ) *token-sequence*
- # *undef identifier*
- # *include <filename>*
- # *include "filename"*
- # *include token-sequence*
- # *line constant "filename"*
- # *line constant*
- # *error token-sequence*<sub>opt</sub>
- # *pragma token-sequence*<sub>opt</sub>
- #
- preprocessor-conditional*

*preprocessor-conditional*:

- if-line text* *elif-parts* *else-part*<sub>opt</sub> # *endif*

*if-line*:

- # *if constant-expression*
- # *ifdef identifier*
- # *ifndef identifier*

*elif-parts*:

- elif-line text*
- elif-parts*<sub>opt</sub>

*elif-line*:

- # *elif constant-expression*

*else-part*:

- else-line text*

*else-line*:

- # *else*